# Contributor: larchunix
# Maintainer: Jelle van der Waa <jelle@archlinux.org>

pkgname=libcomps
pkgver=0.1.20
pkgrel=1
pkgdesc="Comps XML file manipulation library"
arch=('loong64' 'x86_64')
url="https://github.com/rpm-software-management/$pkgname"
license=('GPL2')
depends=('expat' 'libxml2' 'zlib')
makedepends=('cmake' 'python' 'python-setuptools' 'doxygen' 'python-sphinx')
checkdepends=('check')
optdepends=('python: for python bindings')
source=("$url/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('e93299e34da26c14b210f48f324c0c2540515f79f9d748311139f77498317a18')

build() {
  cd "$pkgname-$pkgver"

  cmake -B build -S libcomps \
        -DCMAKE_BUILD_TYPE=None \
        -DCMAKE_C_FLAGS_RELEASE='-DNDEBUG' \
        -DCMAKE_INSTALL_PREFIX=/usr \
        -DCMAKE_INSTALL_LIBDIR=lib \
        -Wno-dev

  make -C build

  make -C build docs
  make -C build pydocs
}

check() {
  cd "$pkgname-$pkgver"

  make -C build test
  make -C build pytest
}

package() {
  cd "$pkgname-$pkgver"

  make -C build DESTDIR="$pkgdir/" install

  mkdir -p "$pkgdir/usr/share/doc/$pkgname"/{doxygen,sphinx}
  cp -Rp build/docs/libcomps-doc/html/ "$pkgdir/usr/share/doc/$pkgname/doxygen/"
  rm -r build/src/python/docs/html/.doctrees
  cp -Rp build/src/python/docs/html/   "$pkgdir/usr/share/doc/$pkgname/sphinx/"

  install -Dp -m644 README.md  "$pkgdir/usr/share/doc/$pkgname/README.md"
}
